
1.1 [bookmark: _Toc133222688]Correction to the Function fl_TC_9_1_2_1_Body
	Function name
	fl_TC_9_1_2_1_Body

	Reason for change
	In the current TTCN, at Step 7 the Security Mode Command message is sent with old security context.
However, in this testcase, a full security context has never been established : In preamble there s a Registration REJ.
According to 24.501 cl 5.4.2.5,
Quote
Upon receipt of the SECURITY MODE REJECT message, the AMF shall stop timer T3560. The AMF shall also abort the ongoing procedure that triggered the initiation of the NAS security mode control procedure.
Both the UE and the AMF shall apply the 5G NAS security context in use before the initiation of the security mode control procedure, if any, to protect the SECURITY MODE REJECT message and any other subsequent messages according to the rules in subclause 4.4.4 and 4.4.5.
Unquote
At present, the SS is not aborting the procedure : it still makes use of the partial security context created during Auth and Ciphering steps before Step 3 in Prose.

This is not in-line with the above quoted 24.501 clause. 


	Summary of change
	It is proposed to perform Authentication & Ciphering before Step 7. New Steps 6A and 6B are introduced for this purpose. 
A Prose CR will be raised for this at RAN5#99.
Also this will not have any impact on TP2, and will make the prose more inline with 24.501.
The TTCN changes made to implement this are highlighted below.

	TTCN module
	SecurityMode_NR5GC.ttcn

	MCC160 Comment
	



    

Before Change:
	function fl_TC_9_1_2_1_Body() runs on NR5GC_PTC
  {
    var NR_CellId_Type v_NGC_NASCellA := f_NR5GC_MapNASCell(ngc_CellA, NG_AllCellsOnSamePLMN);
    var GMM_MobilityInfo_Type v_GMM_MobilityInfo;
    var NR_SRB_COMMON_IND v_ReceivedAsp;
    var NG_NAS_MSG_Indication_Type v_ReceivedMsg;
    var template (value) NG_NAS_DL_Message_Type v_MsgToSend;
    var PLMN_Identity v_ASN_PLMN := f_NR_CellInfo_GetPLMN (v_NGC_NASCellA);
    var NAS_PlmnId v_PLMN := f_NR_Asn2Nas_PlmnId(v_ASN_PLMN);
    var NG_NAS_SecurityParams_Type v_SecurityParams;
    var boolean v_Result;
    var NasCount_Type v_NasCountUL;
    var template (omit) EAP_Message v_EAP;
    var template (omit) ABBA v_ABBA;
    var TrackingAreaCode v_TAC := f_NR_CellInfo_GetTAC (v_NGC_NASCellA);
    var NG_NAS_GutiParameters_Type v_GutiParams := f_NR5GC_CellInfo_GetGutiParameters (v_NGC_NASCellA);
    var template (value) NG_MobileIdentity v_GUTIToSend := f_NG_GutiParameters2MobileIdentity (v_PLMN, v_GutiParams);
    var template (value) NG_TrackingAreaIdList v_TAIList := cs_NG_TAIListNonConsecutive(v_PLMN, {bit2oct(v_TAC)});
    var NG_UE_SecurityCapability v_ChangedCap;

    //@siclog "Step 1" siclog@
    f_UT_SwitchOnUE(UT);
    
    //@siclog "Steps 2-2C" siclog@
    v_ReceivedMsg := f_NR_RRC_ConnEst_DefWithNas (v_NGC_NASCellA, tsc_NR_RRC_TI_Def, ?, tsc_SHT_NoSecurityProtection); // @sic R5-198939, R5s200055, R5s201061 sic@
    //@siclog "Step 2C" siclog@
    if (not f_Check_NG_RegistrationReqMsg (v_GMM_MobilityInfo, v_ReceivedMsg.Pdu, Initial_NoSecurity)) { // @sic R5-202656 sic@
      f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 2C");
    }
    
    //@siclog "Step "2D - 2E" siclog@
    if (px_NAS_5GC_AuthenticationType == AKA_5G) {
      f_NR_Authentication5G_AKA(v_NGC_NASCellA, tsc_SHT_NoSecurityProtection, tsc_SHT_NoSecurityProtection, v_PLMN); // @sic R5-198939, R5-202656 sic@
    } else {
      f_NR_AuthenticationEAP_AKA(v_NGC_NASCellA, tsc_SHT_NoSecurityProtection, tsc_SHT_NoSecurityProtection, v_PLMN); // @sic R5-198939, R5-202656 sic@
    }
    
    v_SecurityParams := f_NR5GC_Security_Get();
    if (px_NAS_5GC_AuthenticationType == AKA_5G) {// only send EAP message and ABBA for for EAP authentication
      v_EAP := omit;
      v_ABBA := omit;
    } else {
      v_EAP := cs_EAP_Message(bit2oct(encvalue(cs_EAP_Success)));
      v_ABBA := cs_ABBA(v_SecurityParams.ABBA);
    }

    f_SS_NG_NAS_ConfigureSecurity(NASCTRL, v_SecurityParams.NAS_Integrity, v_SecurityParams.NAS_Ciphering, true); /* download security keys and reset NAS COUNT */

    v_MsgToSend := f_Get_NG_SecurityModeCmdMsg(v_SecurityParams,
                                               Initial_Secure,
                                               v_GMM_MobilityInfo,
                                               false,
                                               omit,
                                               v_EAP,
                                               v_ABBA);
    v_ChangedCap := fl_InvalidateReplayedSecurityCapabilities(v_MsgToSend.security_Mode_Command.ueSecurityCapability);
    v_MsgToSend.security_Mode_Command.ueSecurityCapability := v_ChangedCap;

    //@siclog "Step 3" siclog@
    SRB.send(cas_NR_SRB_NasPdu_REQ(v_NGC_NASCellA, tsc_NR_RbId_SRB1, -, cs_NG_NAS_Request(tsc_SHT_IntegrityProtected_NewSecurityContext, v_MsgToSend)));

    //@siclog "Step 4" siclog@
    SRB.receive(car_NR_SRB_NasPdu_IND(v_NGC_NASCellA, tsc_NR_RbId_SRB1, cr_NG_NAS_Ind(tsc_SHT_NoSecurityProtection))) -> value v_ReceivedAsp; // @sic R5s200055, R5s201061 sic@
    if (not match (v_ReceivedAsp.Signalling.Nas[0].Pdu.Msg, cr_NG_SECURITY_MODE_REJECT(cr_GMM_GSM_Cause(omit, '00010111'B)))) {
      f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 4");
    } else {
      f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 4");
    }

    //@siclog "Step 5" siclog@
    SRB.send(cas_NR_SRB_NasPdu_REQ(v_NGC_NASCellA, tsc_NR_RbId_SRB1, -, cs_NG_NAS_Request(tsc_SHT_NoSecurityProtection,
                                                                         cs_NG_IDENTITY_REQUEST(tsc_IdType_NG_SUCI))));
    //@siclog "Step 6" siclog@
    SRB.receive(car_NR_SRB_NasPdu_IND(v_NGC_NASCellA,
                                        tsc_NR_RbId_SRB1,
                                        cr_NG_NAS_Ind(tsc_SHT_NoSecurityProtection,   // @sic R5-202656 sic@
                                                          cr_NG_IDENTITY_RESPONSE(?)))) -> value v_ReceivedAsp;
    if (match (v_ReceivedAsp.Signalling.Nas[0].Pdu.Msg.identity_Response.ngMobileId, cr_NG_MobileIdentitySUCI(?, v_ReceivedAsp.Signalling.Nas[0].Pdu.Msg.identity_Response.ngMobileId.otherDigits))) {
      f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 6");
    } else {
      f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 6");
    }

    f_SS_NG_NAS_ConfigureSecurity(NASCTRL, v_SecurityParams.NAS_Integrity, v_SecurityParams.NAS_Ciphering, true); /* @sic R5s190205 sic@ */

    //@siclog "Step 7" siclog@
    v_MsgToSend := f_Get_NG_SecurityModeCmdMsg(v_SecurityParams,
                                               Initial_NoSecurity,  // @sic R5s201061 sic@
                                               v_GMM_MobilityInfo,
                                               true,
                                               omit,
                                               v_EAP,
                                               v_ABBA);
    SRB.send(cas_NR_SRB_NasPdu_REQ(v_NGC_NASCellA, tsc_NR_RbId_SRB1, -, cs_NG_NAS_Request(tsc_SHT_IntegrityProtected_NewSecurityContext, v_MsgToSend)));

    //@siclog "Step 8" siclog@
    SRB.receive(car_NR_SRB_NasPdu_IND(v_NGC_NASCellA, tsc_NR_RbId_SRB1, cr_NG_NAS_IndWithPiggybacking(tsc_SHT_IntegrityProtected_Ciphered_NewSecurityContext))) -> value v_ReceivedAsp;
    v_Result := f_Check_NG_SecurityModeCompleteMsg(v_ReceivedAsp.Signalling.Nas[0].Pdu, Initial_NoSecurity, true); // @sic R5s190435, R5s200055, R5s201061 sic@
    v_NasCountUL := v_ReceivedAsp.Signalling.Nas[0].SecurityProtection.NasCount;
    if (not v_Result) {
      f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 8");
    }
    f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 8");
    
<<SKIPPED CODE>>




After Change:
	<<SKIPPED CODE>>
import from NasEmu_Common4G5G all; //WA#9_1_2_1
import from NR_SecurityFunctions all; //WA#9_1_2_1
.
.
.
function fl_TC_9_1_2_1_Body() runs on NR5GC_PTC
  {
    var NR_CellId_Type v_NGC_NASCellA := f_NR5GC_MapNASCell(ngc_CellA, NG_AllCellsOnSamePLMN);
    var GMM_MobilityInfo_Type v_GMM_MobilityInfo;
    var NR_SRB_COMMON_IND v_ReceivedAsp;
    var NG_NAS_MSG_Indication_Type v_ReceivedMsg;
    var template (value) NG_NAS_DL_Message_Type v_MsgToSend;
    var PLMN_Identity v_ASN_PLMN := f_NR_CellInfo_GetPLMN (v_NGC_NASCellA);
    var NAS_PlmnId v_PLMN := f_NR_Asn2Nas_PlmnId(v_ASN_PLMN);
    var NG_NAS_SecurityParams_Type v_SecurityParams;
    var boolean v_Result;
    var NasCount_Type v_NasCountUL;
    var template (omit) EAP_Message v_EAP;
    var template (omit) ABBA v_ABBA;
    var TrackingAreaCode v_TAC := f_NR_CellInfo_GetTAC (v_NGC_NASCellA);
    var NG_NAS_GutiParameters_Type v_GutiParams := f_NR5GC_CellInfo_GetGutiParameters (v_NGC_NASCellA);
    var template (value) NG_MobileIdentity v_GUTIToSend := f_NG_GutiParameters2MobileIdentity (v_PLMN, v_GutiParams);
    var template (value) NG_TrackingAreaIdList v_TAIList := cs_NG_TAIListNonConsecutive(v_PLMN, {bit2oct(v_TAC)});
    var NG_UE_SecurityCapability v_ChangedCap;   

    var NasCountInfo_Type v_NasCountInfo; //WA#9_1_2_1

    //@siclog "Step 1" siclog@
    f_UT_SwitchOnUE(UT);
    
    //@siclog "Steps 2-2C" siclog@
    v_ReceivedMsg := f_NR_RRC_ConnEst_DefWithNas (v_NGC_NASCellA, tsc_NR_RRC_TI_Def, ?, tsc_SHT_NoSecurityProtection); // @sic R5-198939, R5s200055, R5s201061 sic@
    //@siclog "Step 2C" siclog@
    if (not f_Check_NG_RegistrationReqMsg (v_GMM_MobilityInfo, v_ReceivedMsg.Pdu, Initial_NoSecurity)) { // @sic R5-202656 sic@
      f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 2C");
    }
    
    //@siclog "Step "2D - 2E" siclog@
    if (px_NAS_5GC_AuthenticationType == AKA_5G) {
      f_NR_Authentication5G_AKA(v_NGC_NASCellA, tsc_SHT_NoSecurityProtection, tsc_SHT_NoSecurityProtection, v_PLMN); // @sic R5-198939, R5-202656 sic@
    } else {
      f_NR_AuthenticationEAP_AKA(v_NGC_NASCellA, tsc_SHT_NoSecurityProtection, tsc_SHT_NoSecurityProtection, v_PLMN); // @sic R5-198939, R5-202656 sic@
    }
    
    v_SecurityParams := f_NR5GC_Security_Get();
    if (px_NAS_5GC_AuthenticationType == AKA_5G) {// only send EAP message and ABBA for for EAP authentication
      v_EAP := omit;
      v_ABBA := omit;
    } else {
      v_EAP := cs_EAP_Message(bit2oct(encvalue(cs_EAP_Success)));
      v_ABBA := cs_ABBA(v_SecurityParams.ABBA);
    }

    f_SS_NG_NAS_ConfigureSecurity(NASCTRL, v_SecurityParams.NAS_Integrity, v_SecurityParams.NAS_Ciphering, true); /* download security keys and reset NAS COUNT */

    v_MsgToSend := f_Get_NG_SecurityModeCmdMsg(v_SecurityParams,
                                               Initial_Secure,
                                               v_GMM_MobilityInfo,
                                               false,
                                               omit,
                                               v_EAP,
                                               v_ABBA);
    v_ChangedCap := fl_InvalidateReplayedSecurityCapabilities(v_MsgToSend.security_Mode_Command.ueSecurityCapability);
    v_MsgToSend.security_Mode_Command.ueSecurityCapability := v_ChangedCap;

    //@siclog "Step 3" siclog@
    SRB.send(cas_NR_SRB_NasPdu_REQ(v_NGC_NASCellA, tsc_NR_RbId_SRB1, -, cs_NG_NAS_Request(tsc_SHT_IntegrityProtected_NewSecurityContext, v_MsgToSend)));

    //@siclog "Step 4" siclog@
    SRB.receive(car_NR_SRB_NasPdu_IND(v_NGC_NASCellA, tsc_NR_RbId_SRB1, cr_NG_NAS_Ind(tsc_SHT_NoSecurityProtection))) -> value v_ReceivedAsp; // @sic R5s200055, R5s201061 sic@
    if (not match (v_ReceivedAsp.Signalling.Nas[0].Pdu.Msg, cr_NG_SECURITY_MODE_REJECT(cr_GMM_GSM_Cause(omit, '00010111'B)))) {
      f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 4");
    } else {
      f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 4");
    }

    //@siclog "Step 5" siclog@
    SRB.send(cas_NR_SRB_NasPdu_REQ(v_NGC_NASCellA, tsc_NR_RbId_SRB1, -, cs_NG_NAS_Request(tsc_SHT_NoSecurityProtection,
                                                                         cs_NG_IDENTITY_REQUEST(tsc_IdType_NG_SUCI))));
    //@siclog "Step 6" siclog@
    SRB.receive(car_NR_SRB_NasPdu_IND(v_NGC_NASCellA,
                                        tsc_NR_RbId_SRB1,
                                        cr_NG_NAS_Ind(tsc_SHT_NoSecurityProtection,   // @sic R5-202656 sic@
                                                          cr_NG_IDENTITY_RESPONSE(?)))) -> value v_ReceivedAsp;
    if (match (v_ReceivedAsp.Signalling.Nas[0].Pdu.Msg.identity_Response.ngMobileId, cr_NG_MobileIdentitySUCI(?, v_ReceivedAsp.Signalling.Nas[0].Pdu.Msg.identity_Response.ngMobileId.otherDigits))) {
      f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 6");
    } else {
      f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 6");
    }

    //WA#9_1_2_1 COMMENTED OUT f_SS_NG_NAS_ConfigureSecurity(NASCTRL, v_SecurityParams.NAS_Integrity, v_SecurityParams.NAS_Ciphering, true); /* @sic R5s190205 sic@ */
      
    //WA#9_1_2_1 : Start	
	if (px_NAS_5GC_AuthenticationType == AKA_5G) {
      f_NR_Authentication5G_AKA(v_NGC_NASCellA, tsc_SHT_NoSecurityProtection, tsc_SHT_NoSecurityProtection, v_PLMN); 
    } else {
      f_NR_AuthenticationEAP_AKA(v_NGC_NASCellA, tsc_SHT_NoSecurityProtection, tsc_SHT_NoSecurityProtection, v_PLMN); 
    }
    
    v_SecurityParams := f_NR5GC_Security_Get();
    if (px_NAS_5GC_AuthenticationType == AKA_5G) {
      v_EAP := omit;
      v_ABBA := omit;
    } else {
      v_EAP := cs_EAP_Message(bit2oct(encvalue(cs_EAP_Success)));
      v_ABBA := cs_ABBA(v_SecurityParams.ABBA);
    }
    v_NasCountInfo:= f_SS_NG_NAS_CountGet(NASCTRL); 
    f_SS_NG_NAS_CountSet(NASCTRL,f_NasSecurity_IncrementCOUNT(v_NasCountInfo.DL), f_NasSecurity_IncrementCOUNT(v_NasCountInfo.UL)); 	
	//WA#9_1_2_1 : Stop

    //@siclog "Step 7" siclog@
    v_MsgToSend := f_Get_NG_SecurityModeCmdMsg(v_SecurityParams,
                                               Initial_NoSecurity,  // @sic R5s201061 sic@
                                               v_GMM_MobilityInfo,
                                               true,
                                               omit,
                                               v_EAP,
                                               v_ABBA);
    SRB.send(cas_NR_SRB_NasPdu_REQ(v_NGC_NASCellA, tsc_NR_RbId_SRB1, -, cs_NG_NAS_Request(tsc_SHT_IntegrityProtected_NewSecurityContext, v_MsgToSend)));

    //@siclog "Step 8" siclog@
    SRB.receive(car_NR_SRB_NasPdu_IND(v_NGC_NASCellA, tsc_NR_RbId_SRB1, cr_NG_NAS_IndWithPiggybacking(tsc_SHT_IntegrityProtected_Ciphered_NewSecurityContext))) -> value v_ReceivedAsp;
    v_Result := f_Check_NG_SecurityModeCompleteMsg(v_ReceivedAsp.Signalling.Nas[0].Pdu, Initial_NoSecurity, true); // @sic R5s190435, R5s200055, R5s201061 sic@
    v_NasCountUL := v_ReceivedAsp.Signalling.Nas[0].SecurityProtection.NasCount;
    if (not v_Result) {
      f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 8");
    }
    f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 8");
<<SKIPPED CODE>>



